


Γραφική αναπαράσταση συναρτήσεων μιας μεταβλητής

Γραφική αναπαράσταση της συνάρτησης $\sin(x^2)$ στο διάστημα $0 < x < 4$.

```
x=var('x');
```

```
p=plot(sin(x^2),(x,0,4));p
```



```
3600/2^11.9
```


```
0.941988394807289
```

```
2^30
```

```
1073741824
```


Παράδειγμα:Κοινή γραφική αναπαράσταση της συναρτήσεων $3 \sin(x^2)$ και $e^x - 1$ στο διάστημα $0 < x < 2$.

```
g1=plot(3*sin(x^2),(x,0,2),color="red");
g2=plot(e^x-1,(x,0,2),linestyle="--");
g1+g2
```




Παράδειγμα: Χρήση λογαριθμικού άξονα (εδώ άξονας x)

```
plot(e^(x^2*sin(x)^2), (x,0,6), scale="semilogy")
```



Παράδειγμα: Γραφική αναπαράσταση συνάρτησης με απειρισμούς: Η συνάρτηση $\tan(x)$ στο διάστημα $-2\pi < x < \pi$

```
plot(tan(x), (x,-2*pi,2*pi), ymin=-10, ymax=10, detect_poles=True)
```




Γραφική αναπαράσταση σύνθετων συναρτήσεων

Γραφική αναπαράσταση συνάρτησης

$$f(x) = \begin{cases} x & x < 1 \\ \frac{1}{x^2} & x \geq 1 \end{cases}$$

```
def f(x):
    if x < 1:
        return x
    else:
        return 1/x^2
```

```
plot(f, (x, 0, 10))
```




Γραφική αναπαράσταση καμπυλών σε παραμετρική μορφή

Γραφική αναπαράσταση της καμπύλης

$$\begin{aligned} x(t) &= 2 \cos(t) - \cos(2t) \\ y(t) &= 2 \sin(t) - \sin(2t) \end{aligned}$$

```
t=var('t')
```

```
parametric_plot((2*cos(t)-cos(2*t),2*sin(t)-sin(2*t)),(t,-4,3))
```




```
y=var('y')
```

Γραφική αναπαράσταση της καμπύλης


$$y^2 - x^3 - 2x^2 + x - 1 = 0$$

```
implicit_plot(y^2-x^3-2*x^2+x-1,(x,-5,5),(y,-5,5))
```




Γραφικά σε πολικές συντεταγμένες

```
t=var('t');
polar_plot(1/cosh(t/2),(t,-10,10))
```




Γραφικά τύπου contour

```
y=var('y');
contour_plot(x^2+y^2-x*(y+1), (x,-2,2), (y,-2,2), fill=False, labels=True, axes_labels=['$x$', '$y$'])
```




```
contour_plot(x^2+y^2-x*(y+1), (x,-2,2), (y,-2,2), fill=False, labels=True, axes_labels=['$x$', '$y$'], contours=[0, 5, 10], cmap='hsv')
```




Γραφική αναπαράσταση διανυσματικού πεδίου

```
y=var('y');plot_vector_field((x-y,x),(x,-2,2),(y,-2,2),aspect_ratio=1)
```




```
plot_vector_field((x/sqrt(x^2+y^2),y/sqrt(x^2+y^2)),(x,-2,2),(y,-2,2),aspect_ratio=1,axes=False)
```




Γραφική αναπαράσταση περιοχών

```
region_plot(sin(x*y)+cos(x*y) <= 0, (x, -3, 3), (y, -3, 3), aspect_ratio=1)
```




Μορφοποίηση γραφικών αναπαραστάσεων


```
plot(sin(x)^2, (x, 0, 2*pi), fill='axis')
```




```
plot([sqrt(x), sin(x^2)], (x, 0, 2*pi), fill=[0:1], fillcolor='green')
```



```
plot(x^3*(sin(x^2)), (x, 0, sqrt(8*pi)), gridlines=True, frame=True, axes=False)
```

```
p=plot(sin((pi*x^2-5)/(2*x^2+2)),(x,0,10));
p1=plot(1,(x,0,10),color='red',linestyle='--');
pt=text(r'$f(x)=\sin\left(\frac{x^2-5}{x^2+2}\right)$',(8,0.6),fontsize=20,color='green',ymax=1.2,axes_labels=
['x$', '$f(x)$']);
pta=text('$y=1$',(8,1.05),fontsize=20,color='red');
par=arrow2d((1.3,1.1),(0.55,1.01),color='green',linestyle=':');
pax=text('maximum',(1.5,1.15),fontsize=20,color='green');
p+p1+pt+pta+par+pax
```



Αποθήκευση γραφικών σε αρχείο

```
g=plot(log(abs(sin(x)+2)),(x,-2*pi,2*pi));
g.save('plot.pdf')
```

[plot.pdf](#)

```
g.save('plot.eps')
```


[plot.eps](#)

Γραφική αναπαράσταση δεδομένων

```
data=[round(x,3), round(sin(x^2),3)] for x in srange(0.0,2.6,0.1);
data
```

```
[0.0, 0.0],
[0.1, 0.01],
[0.2, 0.04],
[0.3, 0.09],
[0.4, 0.159],
[0.5, 0.247],
[0.6, 0.352],
[0.7, 0.471],
[0.8, 0.597],
[0.9, 0.724],
[1.0, 0.841],
[1.1, 0.936],
[1.2, 0.991],
[1.3, 0.993],
[1.4, 0.925],
[1.5, 0.778],
[1.6, 0.549],
[1.7, 0.249],
[1.8, -0.098],
[1.9, -0.451],
[2.0, -0.757],
[2.1, -0.955],
[2.2, -0.992],
[2.3, -0.838],
[2.4, -0.5],
[2.5, -0.033]]
```


```
g1=scatter_plot(data);g1
```



```
g1=scatter_plot(data,marker='s',frame=True,axes=False,gridlines=True)
```

Γραφική αναπαράσταση δεδομένων με μπάρες


```
marks = [100,110,120,140,50,180,90,60,20,9,10,2]
g=bar_chart(marks);show(g)
```



Γραφική αναπαράσταση μιγαδικών συναρτήσεων

```
y=var('y')
```

```
p1=complex_plot(sin(y^2), (-3,3), (-3,3));p1
```




Αλληλεπιδραστικά γραφικά

```
v0=8;g=10;
@interact
def voli(u=slider(0.2,89,2)):
    th=u*pi/180;
    html(r'$y=x\tan\theta-\frac{g x^2}{2 v_0^2\cos\theta}$')
    html('Πλάγια βολή θ='+str(round(u,2))+' μοίρες, $v_0$='+str(v0)+'m/s')
    gg=plot(v0*tan(th)*x-g*x^2/(v0^2*cos(th)^2), (x,0,30), ymin=0, ymax=80);
    show(gg)
```

u 0.2000000000000000





$$y = x \tan \theta - \frac{gx^2}{2v_0^2 \cos^2 \theta}$$

Πλάγια βολή $\theta=60.2$ μοίρες, $v_0=8\text{m/s}$




Συνδιασμοί γραφικών

```
t,x,y=var('t x y');
f=1/sqrt((x-1)^2+y^2)-1/sqrt((x+1)^2+y^2)
marks = [100,110,120,140,50,180,90,60,20,9,10,2];
p1=complex_plot(y^3-1/y, (-3,3), (-3,3))
p2=bar_chart(marks)
p3=contour_plot(x^2+y^2-x*y, (x, -2, 2), (y, -2, 2), fill=False,cmap='hsv', labels=True)
p4=polar_plot(1/cosh(t/2), (t,-10,10));
w=graphics_array((p4,p2), (p3,p1));w
```




Γραφική αναπαράσταση συνάρτησης δύο μεταβλητών

```
x,y=var('x,y');plot3d(x^2*y/(x^4 + y^2), (x,-3,3), (y,-3,3),points=500)
```




Παραμετρικά γραφικά στις 3 διαστάσεις

```
u=var('u');parametric_plot3d([sin(u),cos(u),u], (u,0,4*pi))
```



Γραφική αναπαράσταση επιφανειών στις 3 διαστάσεις

```
u, v = var('u,v');
parametric_plot3d((sin(u)*cos(v), cos(u)*cos(v)^2, cos(v)), (u, 0, 2*pi), (v, -pi, pi))
```



Προχωρημένα γραφικά: Χρήση της Matplotlib

```
import matplotlib.pyplot as plt
import numpy as np
dx = 0.01
x = np.arange(0, 3, dx)
f = sin(power(x, 3))
dfdx = 3*power(x, 2)*sin(power(x, 3))
plt.figure(figsize=(5, 5))
plt.plot(x, f, label='f(x)')
plt.plot(x, dfdx, color='red')
plt.savefig('diff.png')
plt.close()
```

